BLOCKSEC

Security Audit

Report for IFTieredSale Contracts

Date: August 30, 2024 **Version:** 1.1

Contact: contact@blocksec.com

Contents

Report Manifest

Item	Description
Client	Impossible Finance
Target	IFTieredSale Contracts

Version History

Version	Date	Description
1.0	July 10, 2024	First release
1.1	August 30, 2024	Add new commits (`Version 3` & `Version 4`)

Signature

Digitally signed by: BlockSec (https://blocksec.com)

SHA1 digest of public key: BAB106D3708BC6DE0963CE05E9478C24B326AE1C

Date: 2024-08-30 10:34:24+0000

Chapter 1 Introduction

1.1 About Target Contracts

Information	Description
Type	Smart Contract
Language	Solidity
Approach	Semi-automatic and manual verification

The focus of this audit is the IFTieredSale Contracts of Impossible Finance [1]. These contracts facilitate tiered node sales with a whitelist mechanism and promotional code discounts. Whitelisted users can make purchases in permitted tiers, with different pricing, owner rewards, and promo code configurations.

Please note that only contracts inside the `contracts` folder in the repository are within the scope of this audit. Other files are not included. Additionally, all dependencies of the smart contract within the audit scope are considered reliable in terms of both functionality and security and are therefore not included in the audit.

The auditing process is iterative. Specifically, we would audit the commits that fix the discovered issues. If there are new issues, we will continue this process. The commit SHA values during the audit are shown in the following table. Our audit report is responsible for the code in the initial version (`Version 1`), as well as new code (in the following versions) to fix issues in the audit report.

Project	Version	Commit Hash
IFTieredSale Contracts	`Version 1`	`17042f18ee4b647787d83b7da7f061328f554b4c`
	`Version 2`	`f96ab456c1a7cae9dbf4de092dd2a32d9d340f18`
	`Version 3`	`4c770b4e8553de8dd86ec8d93b8938b751e0077c`
	`Version 4`	`2b3530c08afadf3ab9c9d54e6d21a9f5a7b69d57`

1.2 Disclaimer

This audit report does not constitute investment advice or a personal recommendation. It does not consider, and should not be interpreted as considering or having any bearing on, the potential economics of a token, token sale or any other product, service or other asset. Any entity should not rely on this report in any way, including for the purpose of making any decisions to buy or sell any token, product, service or other asset.

This audit report is not an endorsement of any particular project or team, and the report does not guarantee the security of any particular project. This audit does not give any warranties on discovering all security issues of the smart contracts, i.e., the evaluation result does not guarantee the nonexistence of any further findings of security issues. As one audit can-

[1] https://github.com/ImpossibleFinance/impossible-node-sale/tree/tiered-sale

not be considered comprehensive, we always recommend proceeding with independent audits and a public bug bounty program to ensure the security of smart contracts.

The scope of this audit is limited to the code mentioned in Section 1.1. Unless explicitly specified, the security of the language itself (e.g., the solidity language), the underlying compiling toolchain and the computing infrastructure are out of the scope.

1.3 Procedure of Auditing

We perform the audit according to the following procedure.

- **Vulnerability Detection** We first scan smart contracts with automatic code analyzers, and then manually verify (reject or confirm) the issues reported by them.
- **Semantic Analysis** We study the business logic of smart contracts and conduct further investigation on the possible vulnerabilities using an automatic fuzzing tool (developed by our research team). We also manually analyze possible attack scenarios with independent auditors to cross-check the result.
- **Recommendation** We provide some useful advice to developers from the perspective of good programming practice, including gas optimization, code style, and etc.

We show the main concrete checkpoints in the following.

1.3.1 Software Security

* Reentrancy
* DoS
* Access control
* Data handling and data flow
* Exception handling
* Untrusted external call and control flow
* Initialization consistency
* Events operation
* Error-prone randomness
* Improper use of the proxy system

1.3.2 DeFi Security

* Semantic consistency
* Functionality consistency
* Permission management
* Business logic
* Token operation
* Emergency mechanism
* Oracle security
* Whitelist and blacklist
* Economic impact
* Batch transfer

1.3.3 NFT Security

* Duplicated item
* Verification of the token receiver
* Off-chain metadata security

1.3.4 Additional Recommendation

* Gas optimization
* Code quality and style

Note *The previous checkpoints are the main ones. We may use more checkpoints during the auditing process according to the functionality of the project.*

1.4 Security Model

To evaluate the risk, we follow the standards or suggestions that are widely adopted by both industry and academy, including OWASP Risk Rating Methodology [2] and Common Weakness Enumeration [3]. The overall *severity* of the risk is determined by *likelihood* and *impact*. Specifically, likelihood is used to estimate how likely a particular vulnerability can be uncovered and exploited by an attacker, while impact is used to measure the consequences of a successful exploit.

In this report, both likelihood and impact are categorized into two ratings, i.e., *high* and *low* respectively, and their combinations are shown in Table 1.1.

Table 1.1: Vulnerability Severity Classification

Impact \ Likelihood	*High*	*Low*
High	High	Medium
Low	Medium	Low

Accordingly, the severity measured in this report are classified into three categories: **High**, **Medium**, **Low**. For the sake of completeness, **Undetermined** is also used to cover circumstances when the risk cannot be well determined.

Furthermore, the status of a discovered item will fall into one of the following four categories:

- **Undetermined** No response yet.
- **Acknowledged** The item has been received by the client, but not confirmed yet.

[2] https://owasp.org/www-community/OWASP_Risk_Rating_Methodology

[3] https://cwe.mitre.org/

- **Confirmed** The item has been recognized by the client, but not fixed yet.
- **Fixed** The item has been confirmed and fixed by the client.

Chapter 2 Findings

In total, we found **two** potential security issues. Besides, we have **six** recommendations and **three** notes.

- Medium Risk: 1
- Low Risk: 1
- Recommendation: 6
- Note: 3

ID	Severity	Description	Category	Status
1	Medium	Incorrect logic for loading `promoCodeAddress` from memory	Software Security	Fixed
2	Low	Incorrect `codePurchaseAmount` accounting	DeFi Security	Fixed
3	-	Apply sanity checks on parameters	Recommendation	Fixed
4	-	Remove redundant code	Recommendation	Fixed
5	-	Correct the typo in the function name	Recommendation	Fixed
6	-	Implement the setter function for `claimRewardsEnabled`	Recommendation	Fixed
7	-	Remove unused variable	Recommendation	Fixed
8	-	Avoid unexpected reverts due to underflows	Recommendation	Fixed
9	-	Potential centralized risks	Note	-
10	-	Lack of support for non-standard ERC20 tokens	Note	-
11	-	Lack of implementation for user withdrawals	Note	-

The details are provided in the following sections.

2.1 Software Security

2.1.1 Incorrect logic for loading `promoCodeAddress` from memory

Severity Medium

Status Fixed in `Version 2`

Introduced by `Version 1`

Description The `IFTieredSale` contract allows users to use an *address promo code* to purchase `saleTokens`. The promo code is considered an *address promo code* when its length equals 42 characters (i.e., the length of an address in hexadecimal format). In such cases, the `promoCodeOwnerAddress` of the promo code is the code itself. However, the contract incorrectly processes these promo codes, resulting in loading of incorrect addresses.

For example, in the `executePurchase` function, it attempts to load the first 20 bytes of the `_promocode` using `mload(add(_promoCode, 20))` on line 273. This approach is problematic as

`mload` directly loads the raw bytes, not the actual characters. As a result, the output is not the correct address but rather truncated UTF-8 bytes.

```
390   function _isAddressPromoCode(string memory _promoCode) internal pure returns (bool) {
391       return bytes(_promoCode).length == 42;
392   }
```

Listing 2.1: contracts/IFTieredSale.sol

```
248   function executePurchase (string memory _tierId, uint256 _amount, uint256 _price, string memory
          _promoCode) private nonReentrant {
249       Tier storage tier = tiers[_tierId];
250       require(!tier.isHalt, "Purchases in this tier are currently halted");
251       require(tier.startTime <= block.timestamp && block.timestamp <= tier.endTime, "Tier is not
              active");
252       require(_amount > 0, "Can only purchase non-zero amounts");
253       require(
254           tier.maxAllocationPerWallet == 0 || purchasedAmountPerTier[_tierId][msg.sender] +
                  _amount <= tier.maxAllocationPerWallet,
255           "Amount exceeds wallet's maximum allocation for this tier"
256       );
257       require(
258           tier.maxTotalPurchasable == 0 || saleTokenPurchasedByTier[_tierId] + _amount <= tier.
                  maxTotalPurchasable,
259           "Amount exceeds tier's maximum total purchasable"
260       );
261
262       totalPaymentReceived += _amount * _price;
263       purchasedAmountPerTier[_tierId][msg.sender] += _amount;
264       saleTokenPurchasedByTier[_tierId] += _amount;
265
266       uint256 totalCost = _amount * _price; // in gwei
267
268       // no need to validate address promo code at purchase
269       if (_isAddressPromoCode(_promoCode)) {
270           if (promoCodes[_promoCode].promoCodeOwnerAddress == address(0)) {
271               address promoCodeAddress;
272               assembly {
273                   promoCodeAddress := mload(add(_promoCode, 20))
274               }
275               promoCodes[_promoCode].promoCodeOwnerAddress = promoCodeAddress;
276           }
277           uint256 ownerRewards = totalCost * addressPromoCodePercentage / 100;
278           totalRewardsUnclaimed += ownerRewards;
279           promoCodes[_promoCode].promoCodeOwnerEarnings += ownerRewards;
280           promoCodes[_promoCode].totalPurchased += totalCost;
281       }
```

Listing 2.2: contracts/IFTieredSale.sol

The same issue is also present in the `_validatePromoCode` function.

```
394   function _validatePromoCode(string memory _promoCode) internal view {
395       require(bytes(_promoCode).length != 0, "Promo code is empty");
```

```
396
397       // if the promo code is an address, check if it has purchased a node code
398       // if the promo code is not an address, check if it is added by the admin (by checking the
              discount percentage)
399       if (!_isAddressPromoCode(_promoCode)) {
400           require(promoCodes[_promoCode].discountPercentage != 0, "Invalid promo code discount
                  percentage");
401           return;
402       }
403
404       // prceed to check if the address has purchased a node
405       address promoCodeAddress;
406       assembly {
407           promoCodeAddress := mload(add(_promoCode, 20))
408       }
```

Listing 2.3: contracts/IFTieredSale.sol

Impact The owner of the address promo code cannot be loaded correctly.

Suggestion Revise the code logic accordingly.

2.2 DeFi Security

2.2.1 Incorrect `codePurchaseAmount` accounting

Severity Low

Status Fixed in `Version 2`

Introduced by `Version 1`

Description In the `IFTieredSale` contract, the `codePurchaseAmount` mapping variable is assigned values incorrectly. According to the code annotation, it should track the total purchased amount. However, in the `whitelistedPurchaseInTierWithCode` function, only the price is counted. Although this variable remains unused in the current contract implementation, the discrepancy may introduce potential issues if its value is used elsewhere.

```
202   function whitelistedPurchaseInTierWithCode(
203       string memory _tierId,
204       uint256 _amount,
205       bytes32[] calldata _merkleProof,
206       string memory _promoCode,
207       uint256 _allocation
208   ) public {
209       // Ensure promo codes are allowed for the tier and the promo code is valid
210       require(tiers[_tierId].allowPromoCode, "Promo code is not allowed for this tier");
211       _validatePromoCode(_promoCode);
212       bytes32 tierWhitelistRootHash = tiers[_tierId].whitelistRootHash;
213       if (tierWhitelistRootHash != bytes32(0)) {
214           require(checkTierWhitelist(_tierId, msg.sender, _merkleProof, _allocation), "Invalid
                  proof");
215           require(purchasedAmountPerTier[_tierId][msg.sender] + _amount <= _allocation, "Purchase
                  exceeds allocation");
```

```
216       }
217
218       uint8 discount = calculateDiscount(_promoCode);
219       uint256 discountedPrice = tiers[_tierId].price * (100 - discount) / 100; // in gwei
220       codePurchaseAmount[_promoCode] += discountedPrice;
221       executePurchase(_tierId, _amount, discountedPrice, _promoCode);
222   }
```

Listing 2.4: contracts/IFTieredSale.sol

Impact The use of an incorrect `codePurchaseAmount` value may lead to unexpected behavior.

Suggestion Revise the `codePurchaseAmount` accounting.

2.3 Additional Recommendation

2.3.1 Apply sanity checks on parameters

Status Fixed in `Version 2`

Introduced by `Version 1`

Description In the `IFTieredSale` contract, several functions lack parameters sanity checks. It is advisable to implement these checks to ensure proper configuration.

1. The `addOperator` function should check that the `operator` address is not a zero address.

```
103       function addOperator(address operator) public onlyRole(DEFAULT_ADMIN_ROLE) {
104           grantRole(OPERATOR_ROLE, operator);
105       }
```

Listing 2.5: contracts/IFTieredSale.sol

2. In the `setTier` function, a check should be added to ensure that the `startTime` and `endTime` of a tier satisfy the following condition:
$block.timestamp < tiers[\_tierId].startTime < tiers[\_tierId].endTime < endTime$.
Additionally, the function emits two identical `TierUpdated` events, which may be redundant.

```
112       function setTier(
113           string memory _tierId,
114           uint256 _price,
115           uint256 _maxTotalPurchasable,
116           uint256 _maxAllocationPerWallet,
117           bytes32 _whitelistRootHash,
118           uint8 _bonusPercentage,
119           bool _isHalt,
120           bool _allowPromoCode,
121           bool _allowWalletPromoCode,
122           uint256 _startTime,
123           uint256 _endTime
124       ) public onlyOperator {
125           // Validate input data
126           require(_bonusPercentage <= 100, "Invalid bonus percentage");
127           require(_price > 0, "Invalid price");
```

```
128         require(_bonusPercentage <= MAX_BONUS_PERCENTAGE, "Invalid bonus percentage");
129
130         tiers[_tierId] = Tier({
131             price: _price,
132             maxTotalPurchasable: _maxTotalPurchasable,
133             maxAllocationPerWallet: _maxAllocationPerWallet,
134             whitelistRootHash: _whitelistRootHash,
135             bonusPercentage: _bonusPercentage,
136             isHalt: _isHalt,
137             allowPromoCode: _allowPromoCode,
138             allowWalletPromoCode: _allowWalletPromoCode,
139             startTime: _startTime,
140             endTime: _endTime
141
142         });
143         emit TierUpdated(_tierId);
```

Listing 2.6: contracts/IFTieredSale.sol

3. The `addPromoCode` function adds a new owner promo code specified by the `_code` parameter. Thus, it should ensure that `_code` is not an empty string and does not match the length of an address to differentiate it from an *address promo code*.

```
157     function addPromoCode(
158         string memory _code,
159         uint8 _discountPercentage,
160         address _promoCodeOwnerAddress,
161         address _masterOwnerAddress,
162         uint8 _baseOwnerPercentageOverride,
163         uint8 _masterOwnerPercentageOverride
164     ) public onlyOperator {
165         if (promoCodes[_code].discountPercentage != 0 || promoCodes[_code].
                promoCodeOwnerAddress != address(0)){
166             revert("Promo code already exists");
167         }
168         // Validate the discount percentage and owner addresses
169         _validatePromoCodeSetting(_discountPercentage, _promoCodeOwnerAddress,
                _masterOwnerAddress, _baseOwnerPercentageOverride,
                _masterOwnerPercentageOverride);
170
171         // Add the promo code
172         promoCodes[_code] = PromoCode({
173             discountPercentage: _discountPercentage,
174             promoCodeOwnerAddress: _promoCodeOwnerAddress,
175             masterOwnerAddress: _masterOwnerAddress,
176             promoCodeOwnerEarnings: 0,
177             masterOwnerEarnings: 0,
178             totalPurchased: 0,
179             baseOwnerPercentageOverride: _baseOwnerPercentageOverride,
180             masterOwnerPercentageOverride: _masterOwnerPercentageOverride
181         });
182         ownerPromoCodes[_promoCodeOwnerAddress].push(_code);
183         ownerPromoCodes[_masterOwnerAddress].push(_code);
184         allPromoCodes.push(_code);
```

```
185         emit PromoCodeAdded(_code, _discountPercentage, _promoCodeOwnerAddress,
                _masterOwnerAddress);
186     }
```

Listing 2.7: contracts/IFTieredSale.sol

Impact The misconfigurations may lead to unexpected behavior.

Suggestion Add sanity checks on the function parameters.

2.3.2 Remove redundant code

Status Fixed in `Version 2`

Introduced by `Version 1`

Description Redundant code exists in the `IFTieredSale` contract and can be safely removed.

1. In the `setTier` function, the check on line 126 is redundant, as the check on line 128 ensures that `_bonusPercentage` is below `MAX_BONUS_PERCENTAGE` (i.e., 5).

```
125     // Validate input data
126     require(_bonusPercentage <= 100, "Invalid bonus percentage");
127     require(_price > 0, "Invalid price");
128     require(_bonusPercentage <= MAX_BONUS_PERCENTAGE, "Invalid bonus percentage");
```

Listing 2.8: contracts/IFTieredSale.sol

2. The `IFTieredSale` contract includes an unused inheritance from `IFWhitelistable`.

```
13      contract IFTieredSale is ReentrancyGuard, AccessControl, IFFundable, IFWhitelistable
            {
14          ...
```

Listing 2.9: contracts/IFTieredSale.sol

3. Within the `Tier` struct, a field named `allowWalletPromoCode` is defined but never used. It can be removed from the struct if it is not reserved for other design purposes.

```
47      struct Tier {
48          uint256 price; // Price per tier in gwei.
49          uint256 maxTotalPurchasable; // Total limit per tier (0 means no limit), specified
                in ether.
50          uint256 maxAllocationPerWallet; // Limit per wallet (0 means no limit), specified
                in ether.
51          uint8 bonusPercentage; // Additional bonus percentage applicable for this tier.
52          bytes32 whitelistRootHash; // Merkle root hash for whitelisting.
53          bool isHalt; // Flag to halt transactions for this tier if set to true.
54          bool allowPromoCode; // Flag to allow promo codes for this tier.
55          bool allowWalletPromoCode; // Flag to allow promo codes specific to wallets.
56          uint256 startTime; // Start time for this tier.
57          uint256 endTime; // End time for this tier.
58      }
```

Listing 2.10: contracts/IFTieredSale.sol

4. In the `withdrawPromoCodelRewards` function, the non-zero length check on line 344 is redundant, as the same check is performed in the `_validatePromoCode` function.

```
342      function withdrawPromoCodelRewards (string memory _promoCode) public nonReentrant {
343          require(claimRewardsEnabled, "Claim rewards is disabled");
344          require(bytes(_promoCode).length > 0, "Invalid promo code");
345          _validatePromoCode(_promoCode);
```

Listing 2.11: contracts/IFTieredSale.sol

Impact N/A

Suggestion Remove redundant codes from the `IFTieredSale` contract.

2.3.3 Correct the typo in the function name

Status Fixed in `Version 2`

Introduced by `Version 1`

Description There is a function named `withdrawPromoCodelRewards` in the `IFTieredSale` contract, which appears to be a potential typo.

Impact N/A

Suggestion Revise the function name as `withdrawPromoCodeRewards`.

2.3.4 Implement the setter function for `claimRewardsEnabled`

Status Fixed in `Version 2`

Introduced by `Version 1`

Description The `claimRewardsEnabled` variable in the `IFTieredSale` contract is used in the `withdrawReferenceRewards` and `withdrawPromoCodelRewards` functions to determine if reward withdrawal is enabled. However, this variable is initialized as true and does not have a public setter function in the contract to update its value.

```
40   bool public claimRewardsEnabled = true;
```

Listing 2.12: contracts/IFTieredSale.sol

```
314  function withdrawReferenceRewards () public nonReentrant {
315      address promoCodeOwner = msg.sender;
316      require(claimRewardsEnabled, "Claim rewards is disabled");
```

Listing 2.13: contracts/IFTieredSale.sol

```
342  function withdrawPromoCodelRewards (string memory _promoCode) public nonReentrant {
343      require(claimRewardsEnabled, "Claim rewards is disabled");
344      require(bytes(_promoCode).length > 0, "Invalid promo code");
345      _validatePromoCode(_promoCode);
```

Listing 2.14: contracts/IFTieredSale.sol

Impact The reward withdrawals are always enabled, rendering the function checks redundant.

Suggestion Add a privileged setter function to update the `claimRewardsEnabled` variable, or remove all related code.

2.3.5 Remove unused variable

Status Fixed in `Version 4`

Introduced by `Version 3`

Description In the `validateWalletPromoCode` function, the `sum` variable is updated within loops but never used. To optimize gas, this variable can be safely removed.

```
441   function validateWalletPromoCode(address promoCodeAddress) public view returns (bool) {
442       if (promoCodeAddress == address(0)) {
443           return false;
444       }
445
446       uint256 sum = 0;
447       for (uint i = 0; i < tierIds.length; i++) {
448           if (tiers[tierIds[i]].price == 0) {
449               continue;
450           }
451           if (purchasedAmountPerTier[tierIds[i]][promoCodeAddress] > 0) {
452               // return true if the address has purchased at least one node
453               sum += purchasedAmountPerTier[tierIds[i]][promoCodeAddress];
454               return true;
455           }
456       }
457       return false;
458   }
```

Listing 2.15: contracts/IFTieredSale.sol

Impact N/A

Suggestion Remove the unused variable.

2.3.6 Avoid unexpected reverts due to underflows

Status Fixed in `Version 4`

Introduced by `Version 3`

Description In the `getAllPromoCodeInfo` function, the requirement `fromIdx < toIdx` should be moved after resetting `toIdx` to `allPromoCodes.length`. Failing to do so could cause an unexpected revert due to underflow in cases where the input parameters meet the condition `allPromoCodes.length < fromIdx < toIdx`. Specifically, after `toIdx` is reset on line 576, it may becomes less than `fromIdx`, leading to a function revert due to underflow rather than returning a formatted error message. The same issue is present in the `getAllPromoCodes` function.

```
557   function getAllPromoCodeInfo(uint256 fromIdx, uint256 toIdx) public view returns (PromoCode[]
          memory) {
558       require(fromIdx < toIdx, "Invalid range");
559       if (toIdx > allPromoCodes.length) {
560           toIdx = allPromoCodes.length;
561       }
562       PromoCode[] memory promoCodeInfos = new PromoCode[](toIdx - fromIdx);
563       for (uint i = fromIdx; i < toIdx; i++) {
564           promoCodeInfos[i - fromIdx] = promoCodes[allPromoCodes[i]];
565       }
566       return promoCodeInfos;
567   }
```

Listing 2.16: contracts/IFTieredSale.sol

```
573   function getAllPromoCodes(uint256 fromIdx, uint256 toIdx) public view returns (string[] memory)
          {
574       require(fromIdx < toIdx, "Invalid range");
575       if (toIdx > allPromoCodes.length) {
576           toIdx = allPromoCodes.length;
577       }
578       string[] memory promoCodeList = new string[](toIdx - fromIdx);
579       for (uint i = fromIdx; i < toIdx; i++) {
580           promoCodeList[i] = allPromoCodes[i];
581       }
582       return promoCodeList;
583   }
```

Listing 2.17: contracts/IFTieredSale.sol

Impact These functions may revert without a formatted error message in certain cases.

Suggestion Revise the code accordingly.

2.4 Note

2.4.1 Potential centralized risks

Introduced by `Version 1`

Description In the `IFTieredSale` contract, there are privileged roles capable of modifying critical configurations, such as adding or removing operators and changing the time of an active sale tier. This introduces a risk of centralization. If the private keys of these privileged accounts are leaked, the launchpad could potentially be compromised.

2.4.2 Lack of support for non-standard ERC20 tokens

Introduced by `Version 1`

Description The `IFTieredSale` contract should only support standard ERC20 tokens. Supporting non-standard ERC20 tokens (e.g., deflationary tokens) can introduce potential security

risks. For instance, with deflationary tokens, the actual amount received by users may differ from their expectations.

2.4.3 Lack of implementation for user withdrawals

Introduced by `Version 1`

Description The `IFTieredSale` contract inherits from `IFFundable`, which supports funding the contract with sale tokens. Users should be allowed to withdraw their purchased sale tokens after the sale ends. However, the contract does not implement an external withdraw function to facilitate user withdrawals.

Feedback from the Project We don't plan to implement this for now since it is a node sale. We'll airdrop the nodes accordingly, so the users don't need the sale token.


BOOST WEB3 THROUGH NEXT-GENERATION SECURITY & USABILITY INNOVATIONS